PAIA T.M.

CS-87

CASSETTE INTERFACE

ASSEMBLY

AND

USING MANUAL

CS-87 PARTS LIST

Prior to beginning assembly, check the supplied parts against the following parts list.

QUANTITY	
1	Ceramic transducer disc
1	8:500 Transformer
1	.1 mfd. mylar capacitor
1	16 pin IC socket
1	8 pin IC socket
1	1702A (POTSHOT) PROM
1	4042 Integrated Circuit
1	24 pin IC socket
1	8 pin header
1	TIL-209 LED
1	27K resistor (red-violet-orange)
1	3.3K resistor (orange-orange-red)
1	12 inch length of 4 conductor ribbon cable
2	1K resistor (brown-black-red)
2	10K resistor (brown-black-orange)
2	100K resistor (brown-black-yellow)
2	2N5139 transistors
2	miniature phone plug

If you should find any parts missing, damaged, or otherwise unusable, contact PAIA Electronics, Tech. Services for replacement. Along with your request, we must have the packer number from the parts bag, and the order number under which this kit was ordered.

The parts that constitute the CS-87 cassette interface option fit entirely on the 8700 Computer/Controller board.

CS-87 ASSEMBLY

Install the following components on a PROPERLY OPERATING 8700 Computer Controller circuit board. DO NOT proceed with the assembly of this option until the 8700 has been fully verified as being operational. Remove all power and peripheral connectors from the 8700 before proceeding with the CS-87 installation.

All of the DOs and DON'Ts that were mentioned in the assembly manual for the 8700 apply here also.

Begin assembly by installing the following resistors:



Silver or gold - disregard this band.

PART NUMBER	VALUE	COLOR CODE A-B-C
( ) R4	3300 ohms	orange-orange-red
( ) R46, R47 (2 parts)	100K	brown-black-yellow
( ) R48	1000 ohms	brown-black-red
( ) R49, R50 (2 parts)	10K	brown-black-orange
( ) R51	1000 ohms	brown-black-red
( ) R52	27K	red-violet-orange

Install the single ceramic disk (or in some kits, mylar) capacitor



( ) C10 ..... 0.1 mfd.

There are two 2N5139 transistors which are used for relay drivers in conjunction with the POTSHOT Cassette system tape motion control software. Install these transistors now.

( ) Q8, Q9 (2 parts) 2N5139

The single Light Emitting Diode provides an indication that data is coming into the computer from the cassette player. Orientation of the LED is keyed by the length of the two leads, the short lead is the cathode and should be installed in the circuit board hole marked -. When installing the LED, leave it as high above the surface of the board as possible so that it will be visible above the displays in the cut-out provided in the 8700 Keyboard circuit board. When installing, leave the lead of the LED as long as possible for maximum visibility.



( ) D2 ..... TIL-209 Light Emitting Diode

The 500:8 interstage transformer provides coupling between the cassette recorders output and the computer's input. NOTE that while a solder pad is provided for the center tap on the primary side of this transformer, there are no connections to the pad and this lead may be clipped off if desired. NOTE ALSO that some transformers may have three leads coming from both sides. On these devices, the primary side of the transformer is marked with a "P" (which on some transformers comes out nothing more than a black smudge). The leads coming from this side of the transformer are to be installed in the circuit board holes closest to the edge of the board and the center leads may be clipped from both sides of these transformers.

( ) T1 ..... 500:8 interstage transformer

Install the following sockets at the positions indicated. As with the 8700, observe the polarity notches of the sockets.

( ) 24 pin socket at IC 18

( ) 16 pin socket at IC 20

( ) 8 pin socket at J9

Install the 2 IC's in their sockets. WARNING CMOS PARTS. Leave these parts in their conductive foam carrier until ready to install, do not wear synthetic clothing during installation. Observe orientation notches during installation.

( ) IC 18..... 1702A POT-SHOT Cassette PROM
( ) IC 20..... 4042 quad latch

A piezo-electric transducer is part of the CS-87, which produces an audible "beep" providing positive indication of keyboard activations. The software required to drive the beeper is included in the PIEBUG monitor program. THE BEEPER IS NOT REQUIRED FOR POT-SHOT OPERATION.

Depending on the final installation of the 8700 system, the beeper may be mounted either on the 8700 board as illustrated shortly, or, for maximum volume, may be mounted on a sounding surface external to the computer.

If you elect to mount the beeper on the 8700 board, proceed as follows:

( ) Solder the wire connected to the center silver colored disc of the transducer to the top end of resistor R46 as shown in figure 2 . Be careful not to short adjacent conductors during soldering.



FIGURE 2

( ) Solder the wire coming from the copper outer disk of the transducer to the ground conductor at the bottom of C10.

( ) Thoroughly clean the back of the beeper and the top of the transformer T1 and using contact cement or other adhesive mount the beeper to the transformer as shown in figure 3 . NOTE that the beeper mounts off-center on the transformer.



FIGURE 3

THIS COMPLETES INSTALLATION OF COMPONENTS ON THE 8700 CIRCUIT BOARD. We are now ready to prepare connecting cables.

The header to which the cables connect is somewhat fragile and can also be damaged by excessive soldering temperatures.

Because of the variety of configurations and user-options available, pre-assembled cables are not supplied for the cassette interface.

While wires are being soldered to the header, some sort of vise must be used to serve both as heat sink and to hold the pins rigid. If a vise with small enough jaws is not available, one may be rigged from a pair of needle-nosed pliers held closed by wrapping a rubber band around the handles. PROVIDING SOME SORT OF MECHANICAL SUPPORT AND HEAT SINK IS ESSENTIAL. Do not attempt the following assembly steps without this.

ALSO, proper preparation of both wires and header will simplify assembly. Pre-tin the header pins before soldering the wire in place (note that only the "U" shaped part of the pin to which the wire will attach need be tinned - do not tin the portion of the pin which will mate with the socket).

Prepare the ends of the wire which are to be soldered to the header pin by stripping approximately 1/4" of insulation from its end, tightly twisting the exposed strands, tinning by heating the strands and flowing a small amount of solder into them, and, finally, by clipping the now-tinned exposed wire so that only 1/8" extends beyond the end of the insulation.

( ) Prepare one end of the 4 conductor ribbon cable by first separating the four conductors into two pairs. Cut 1/4" from the end of one of the two pairs. Separate into two pairs of conductors and remove 1/4" from the length of one pair. Separate the ribbon into four conductors and strip 1/4" of insulation from the end of each of the leads. Tightly twist and tin the exposed conductors. Finally, clip the 4 exposed, tinned conductors so that only 1/16" extends beyond the ends of the insulation. Snip



Figure 4



Figure 5

1/4" of insulation from the end of each of the conductors. Twist tightly and tin. After tinning, cut so that 1/16" of twisted conductor extends beyond insulation.



Figure 6

( ) While pins 1-4 of the cassette connector header are being held in a vise, solder the two shorter leads prepared above to pins 3 and 4 of the header.

( ) Similiarly, while holding pins 5-8 in a vise, solder the longer leads prepared above to pins 5 and 6.



*LENGTH AS REQUIRED

Figure 7

The length of ribbon cable supplied with the CS-87 option is intentionally longer than will be required for most applications. Trim the length of the cable as required before installing the miniature phone jacks.

( ) Separate the free end of the ribbon cable into two pairs of conductors. About 2 inches of separation will be sufficient in most cases. Further separate the two pairs into individual conductors (about 1/2" separation) and strip 1/4" of insulation from each end wire before twisting and tinning the exposed strands.

( ) Locate the two miniature phone plugs supplied, unscrew the cover from one and slip the cover over the end of the pair of wires coming from header pins 3 and 4, (the threaded end of the cover should be toward the free end of the cable.) Solder one of the two wires to the short soldering lug of the plug and the other to the ground frame. (Polarity is not important here). Screw the plug cover back in place. This plug should be labeled "EAR".

( ) Unscrew the cover of the second phone jack and slide it over the remaining two conductors of the cable. Solder the wire coming from pin 6 of the header to the soldering lug of the phone plug and the wire coming from pin 5 of the header to the ground frame of the plug. OBSERVE POLARITY. Screw the plug cover back in place. This plug should be labeled "mic".

( ) Mate the cassette connector header with the socket J9 on the 8700 board (observe orientation - if assembled as above, the cables will naturally exit the connector over the nearest edge of the 8700 board).

THIS COMPLETES ASSEMBLY OF THE CS-87 CASSETTE INTERFACE SYSTEM. Re-connect the power supply and keyboard for an initial testing sequence to follow, but do not yet totally re-assemble the computer.

Testing and Familiarization

RECORDER SELECTION

The CS-87 "POT-SHOT" cassette option has been carefully designed to reliably save and retrieve programs and data using extremely inexpensive, low-fidelity, audio recording equipment. While the Phillips format cassettes have proven to be exceptionally convenient because of their size and ease of storing and handling, any recording scheme (even reel-to-reel) can be used with equally predictable results, though perhaps more hassle.

The only hard and fast requirements of the recorder selected are that it have ready access to external input sources (mic and/or aux. inputs) and some provision for a high level output (ear and/or ext. spkr. jacks). The presence or absence of an Automatic Level Control circuit is immaterial to the operation of the system. A "remote" input jack capable of starting and stopping the recorder will be handy if you think you may be adding the optional tape motion control relays at a later date.

We highly recommend the Realistic CTR-34 recorder sold by Radio-Shack for a number of reasons including low cost, convenient size and ready availability.

TAPE SELECTION

Consideration should be given to the kind of tape you will use for storing digital programs. A problem known to magnetic tape media as "dropout" can scrub your programs. Dropout causes the signal on tape to fade away for very brief periods of time and is caused by fluctuations in the thickness of the iron oxide coating on the tape. This is a problem that can go unnoticed when recording voice or music but the loss of one bit of data is fatal when trying to load a program into your computer. Therefore you should find a brand of tape that is reliable and stick with it. Also dropout is always a problem with any tape at the very front of the tape where it is connected to the leader. Avoid this area by skipping the first 15 or 30 seconds of the tape before saving the first program on that cassette.

TESTING

Connect the cassette connectors "mic" plug to your recorders "mic" (first choice) or "aux" (second choice) input jacks and the "ear" plug to the "ear" or "ext spkr" jacks. Apply power to the 8700 Computer/Controller and verify that nothing has happened during installation of the interface system to change the normal operating characteristics of the processor. Since we will be initially testing the system by storing and then retrieving the counting demo program from the 8700 Assembly and Using Manual, entry and execution of this program is a reasonable vehicle for accomplishing this initial important test. Load this program from the keyboard and verity that it runs properly.

DEMO COUNTING PROGRAM

ADDR	CODE	LABEL	INSTRUCTION	COMMENTS
0000	A9 00	BEGIN	LDA #0	;CLEAR ACCUMULATOR
0002	8D 20 08	REPEAT	STA $0820	;DISPLAY ACC
0005	A0 00		LDY #0	;CLR Y
0007	A2 50		LDX #$50	;SPEED SETTING (IN HEX)
0009	C8	LOOP	INY	;DELAY LOOP
000A	D0 FD		BNE LOOP	;BRANCH UNTIL Y=0
000C	CA		DEX	;CHECK SPEED
000D	D0 FA		BNE LOOP	;BRANCH UNTIL X=0
000F	F8		SED	;SET DECIMAL MODE
0010	18		CLC	;CLR CARRY
0011	69 01		ADC #1	;ADD 1 TO ACC
0013	4C 02 00		JMP REPEAT	;DO IT ALL AGAIN

SAVING AND LOADING PROGRAMS

Before performing tape operations, there is some information that the computer needs to successfully complete these tasks:

1) The beginning address of the program or data to be saved or loaded. (4 hexadecimal digits; two bytes)
2) The ending address of the block of data that is to be saved or loaded. (4 hexadecimal digits; two bytes)
3) A file identifier number (2 hex digits; one byte)
4) A one byte (2 hex digits) indicator for either a load operation (11) or a dump operation (DD).

All of this information is entered as a continuous string of digits from the processor keyboard. THE ENTER KEY IS NEVER USED WHILE KEYING IN THIS INFORMATION.

Using the counting program (which you should have entered by now) as an example; this program begins at location $0000 and ends at location $0015 (the second byte of the operand for the JUMP at location $13). Our first 8 entries from the keyboard will, therefore, be;

BEG. ADDR. END ADDR.

0-0-0-0-0-0-1-5 display shows: 15

Any file identifier number may be assigned to this file (except 00 which has special significance as we will see shortly). Since this is our first file on the tape, we will assign it number 01. Your next two keystrokes will be;

ID

0-1 display shows: 01

The final entry will be the LOAD/DUMP "switch". In this case, for a dump from the computer to the tape recorder, DD (think of the D as meaning Dump). Your next two keystrokes

display shows: dd

If you made any mistakes while keying in the preceding information you have no choice but to begin again and key in the whole sequence;



When the preceding data has been entered correctly, you are ready to perform the dump. Put your recorder in the "record" mode and start the tape rolling. Allow about a ten second interval (see TAPE SELECTION) and press the 8700's TAPE control key.

A number of things will happen now. Simultaneously the display should clear to 00 and the beeper should begin to produce a constant tone (which is the leader and synchronization bits). After about three seconds of this tone, its sound should change slightly (as data begins to flow out) and the displays should begin counting, showing the low byte of the address of the data currently being transferred.

It will only take a second or two for the program to dump and when finished the displays will once again clear to 00 and the tone from the beeper should stop.

The program is now stored on tape and if you wish you may rewind the tape, put the recorder in "play" mode and listen to the tone that was recorded on the tape (it should sound very similar to what you heard through the beeper). Be sure to unplug the "ear" jack if you want to listen to the tape - and plug it back in when ready to load a program.

THE BIG TEST

Now we are ready to find out if it worked. Wipe out the program in the memory of the computer by temporarily disrupting power. Restore power and RESET.

Before loading the program for the first time, the playback level of the recorder must be set. With many tape interfaces this is a hit-or-miss proposition, but POT-SHOT provides the level indicating LED D2 (directly above the displays) for this purpose.

Turn your recorders volume control up fully and begin playing the tape. When the program is found the level-LED should begin to glow brightly. (residual clicks and pops on the tape may cause the LED to wink briefly - this should not affect the operation of the system). When sustained glow is observed, quickly (since you have only a few seconds of actual material recorded on tape) reduce the level of the volume control until you see a slight decrease in the brightness of the LED. This should be the proper playback level.

Loading programs follows the same format as did the dump with a continuous string of keyboard entries indicating beginning address, ending address, ID number and LOAD/ DUMP switch.

The first 8 entries, beginning and ending address;

BEG. ADDR. END ADDR.

0-0-0-0-0-0-1-5 display shows: 15

The next two entries, ID#;

0-1 display shows: 01

and finally the LOAD "switch"; think of the 1's as L's - as in LOAD.

LS

1-1 display shows: 11

As with the dumping procedure, if you make any mistakes in entering this data you have no alternative but to begin again and enter the entire sequence;

BEG. ADDR. END ADDR. ID LOAD

0-0-0-0-0-0-1-5 -0-1-1-1

Rewind, put your recorder in the "play mode, start the tape rolling and touch the 8700's TAPE key. You should immediately hear an artoo-type "bleep" from the beeper (it will sound different from the normal key-stroke beep) and the display should clear to 00.

When the leader and synch tone of the information recorded on tape is found, the single level LED will begin to glow continuously and after a few seconds the displays will begin to count. The counting will continue for a few seconds after which one of three things will happen:

1) The level LED will extinguish, the beeper will beep and the displays will show "AA". If this happens you're in good shape - the AA indicates "A-OK" , a good load. You may now verify that the program is back by keying in its beginning address and RUNning it.

-otherwise-

2) The level LED will extinguish, the beeper will sound and the displays will show EE, an Error. This is a check sum error and it may mean that a transient (probably on the tape itself) prevented the data being entered back into the computer properly. You may try varying the volume control setting of the recorder and trying again.
NOTE - when POT-SHOT announces an error it is in most cases not necessary to enter all of the addresses, identifiers, etc. again. These are saved in the memory of the computer and it is only necessary to re-wind the tape and try again by touching the TAPE key. The exception is when any other keyboard entries other than TAPE are made before trying to load again.

3) A third possibility is that the level LED will extinguish and everything will simply stop with no beep and the displays showing some apparently random number. This indicates that the computer does not realize that it should be finished and is still looking for more data. This is a strong indication that improper keyboard entries were made either during preparation for the dump or the load. Try again and if still no luck try Dumping the program again (which at this point must unfortunately be loaded from the keyboard).

SUMMARY

LOAD

Entry Sequence:



1) Keyboard entries of beginning and ending addresses must each be 4 digits, short addresses not allowed.

2) ENTER key not used during entry sequence.

3) ØØ identifier causes first file encountered to be loaded.

4) AA - displayed at end of load indicates OK
EE - displayed at end of load indicates check sum error.

5) If you have a file on tape of which the end address is not known, it can be determined with the following method:

   01) Load the file as normal but with an end address of "FFFF".
   02) Hit RESET when the displays stop counting.
   03) Examine the pointer with the pointer keys (PCH, PCL).
   04) Subtract 2 from the pointer (in HEX) and that will be the end address.

DUMP

Entry Sequence:



1) Keyboard entries of beginning and ending addresses must each be 4 digits, short addresses not allowed.

2) ENTER key not used during entry sequence.

3) ØØ identifier NOT ALLOWED.

TAPE MOTION CONTROL

(if used)

Manual control sequence:

0-0-TAPE

Recorder will remain activated until next keyboard entry.

ZERO PAGE LOCATIONS

PIEBUG BUFFER

F0	DUMP/LOAD/MAN SWITCH (tested for zero, bit 7)
F1	IDENTIFIER
F2	LSB END ADDR.
F3	MSB END ADDR.
F4	LSB BEG ADDR.
F5	MSB BEG ADDR.
F6	LSB POINTER
F7	MSB POINTER

POT-SHOT
Monitor Listing

```
0150  0200                ;
0160  0200                ;
0170  0200                ;
0180  0200                ;  *********************************
0190  0200                ;  *                               *
0200  0200                ;  *  POT-SHOT CASSETTE SYSTEM     *
0210  0200                ;  *  WRITTEN BY ROGER WALTON      *
0220  0200                ;  *  COPYRIGHT 1977 BY PAIA       *
0230  0200                ;  *    ELECTRONICS, INC.          *
0240  0200                ;  *  VERSION 1.0                  *
0250  0200                ;  *                               *
0260  0200                ;  *********************************
0270  0200                ;
0280  0200                PORT    =$0900          ;CASSETTE I/O PORT
0290  0200                DISPLY  =$0820          ;LED DISPLAY
0300  0200                PIEBUG  =$0F52          ;DISP ACC; GOTO PIEBUG
0310  0200                BEEP    =$0F22          ;BEEP SUB IN PIEBUG
0320  0200                ;
0330  0200                STATUS  =$EF            ;INPUT BIT STATUS
0340  0200                CHKSUM  =$EE            ;CHECKSUM
0350  0200                PNTER   =$F6            ;16 BIT ADDR POINTER
0360  0200                COMAND  =$F0            ;LOAD/DUMP COMMAND
0370  0200                IDENT   =$F1            ;FILE IDENTIFIER
0380  0200                ENDADR  =$F2            ;END ADDR
0390  0200                BEGADR  =$F4            ;BEGINNING ADDR
0400  0200                ;
0410  0200                        *=$0E00
0420  0E00                ;
0430  0E00  20 25 0E      START   JSR SNDBIT      ;START TAPE
0440  0E03  8C 20 08              STY DISPLY      ;CLEAR DISPLAY
0450  0E06  A5 F4                 LDA BEGADR      ;MOVE BEGINNING
0460  0E08  85 F6                 STA PNTER       ;  ADDR TO POINTER
0470  0E0A  A5 F5                 LDA BEGADR+1
0480  0E0C  85 F7                 STA PNTER+1
0490  0E0E  A5 F0                 LDA COMAND      ;GET COMMAND
0500  0E10  F0 07                 BEQ MANUAL      ;BR IF COMMAND= "00"
0510  0E12  20 AA 0E              JSR CASS        ;DUMP OR LOAD BLOCK
0520  0E15  18                    CLC
0530  0E16  20 22 0F              JSR BEEP        ;STOP TAPE AND BEEP
0540  0E19  4C 52 0F      MANUAL  JMP PIEBUG      ;RETURN TO MONITOR
0550  0E1C                ;
0560  0E1C                ;
0570  0E1C                ;
0580  0E1C                ;     DELAY SUBROUTINE
0590  0E1C                ;     THIS SUB DELAYS FOR ONE HALF CYCLE
0600  0E1C                ;     (2000 HZ).  Y IS CLEARED, X, A, AND
0610  0E1C                ;     CARRY ARE PRESERVED.
0620  0E1C                ;
```

```
0630  0E1C  8D 00 09    DELAY  STA PORT        ;UPDATE TONE
0640  0E1F  A0 0F               LDY #15
0650  0E21  88          DLY    DEY
0660  0E22  D0 FD               BNE DLY         ;BR UNTIL DELAY FINISHED
0670  0E24  60                 RTS             ;RETURN
0680  0E25             ;
0690  0E25             ;
0700  0E25             ;
0710  0E25             ;      SEND BIT SUBROUTINE
0720  0E25             ;      THIS SUB SENDS THE CARRY BIT TO THE TAPE.
0730  0E25             ;      A "1" BIT CONSISTS OF 16 CYCLES OF 2000 HZ
0740  0E25             ;      AND A "0" BIT CONSISTS OF 8.  THIS SUB
0750  0E25             ;      TURNS RELAY 1 ON AND RELAY 2 OFF.  Y IS
0760  0E25             ;      CLEARED, X, A, AND CARRY ARE PRESERVED.
0770  0E25             ;
0780  0E25  48          SNDBIT PHA             ;SAVE A
0790  0E26  8A                 TXA
0800  0E27  48                 PHA             ;SAVE X
0810  0E28  A2 10              LDX #16         ;NO. OF CYCLES FOR A "1"
0820  0E2A  B0 02              BCS CYCLE       ;BRANCH IF "1" BIT
0830  0E2C  A2 08              LDX #8          ;NO. OF CYCLES FOR A "0"
0840  0E2E  A9 B0       CYCLE  LDA #%10110000  ;TAPE ON; OUTPUT HIGH
0850  0E30  20 1C 0E           JSR DELAY       ;UPDATE PORT AND DELAY
0860  0E33  A9 80              LDA #%10000000  ;TAPE ON; OUTPUT LOW
0870  0E35  20 1C 0E           JSR DELAY       ;UPDATE PORT AND DELAY
0880  0E38  CA                 DEX             ;LAST CYCLE?
0890  0E39  D0 F3              BNE CYCLE       ;BRANCH IF NOT
0900  0E3B  A9 90              LDA #%10010000  ;TAPE ON; OUTPUT NEUTRAL
0910  0E3D  A2 10              LDX #16         ;DELAY COUNTER
0920  0E3F  20 1C 0E    GAP    JSR DELAY       ;PRODUCE A GAP
0930  0E42  CA                 DEX             ;DELAY FINISHED?
0940  0E43  D0 FA              BNE GAP         ;BRANCH IF NOT
0950  0E45  68                 PLA
0960  0E46  AA                 TAX             ;RESTORE X
0970  0E47  68                 PLA             ;RESTORE A
0980  0E48  60                 RTS             ;RETURN
0990  0E49             ;
1000  0E49             ;
1010  0E49             ;
1020  0E49             ;      DETECT BIT SUBROUTINE
1030  0E49             ;      THIS SUB WILL PICK UP ONE BIT FROM
1040  0E49             ;      THE TAPE AND RETURN WITH IT IN THE
1050  0E49             ;      CARRY FLAG.  Y IS CLOBBERED, X AND A
1060  0E49             ;      ARE PRESERVED.  RELAYS ARE NOT AFFECTED.
1070  0E49             ;
1080  0E49  48          DETBIT PHA             ;SAVE A
1090  0E4A  8A                 TXA
1100  0E4B  48                 PHA             ;SAVE X
1110  0E4C  AD 00 09    TONE   LDA PORT        ;LOOK FOR START OF TONE
1120  0E4F  10 FB              BPL TONE        ;BRANCH UNTIL FOUND
1130  0E51  A2 00              LDX #0          ;CLEAR COUNTER
1140  0E53  E8          COUNT  INX             ;COUNT TRANSITIONS
1150  0E54  A0 23              LDY #35         ;TIME LIMIT
1160  0E56  AD 00 09           LDA PORT        ;CHECK INPUT
1170  0E59  85 EF              STA STATUS      ;SAVE INPUT STATUS
1180  0E5B  AD 00 09    CHECK  LDA PORT        ;CHECK INPUT
```

```
1190  0E5E  45 EF           EOR STATUS      ;HAS IT SWITCHED?
1200  0E60  30 F1           BMI COUNT       ;IF SO, BRANCH
1210  0E62  88              DEY             ;TIME UP?
1220  0E63  D0 F6           BNE CHECK       ;IF NOT, BRANCH
1230  0E65  E0 08           CPX #8          ;DOES TONE BURST QUALIFY?
1240  0E67  90 E3           BCC TONE        ;BRANCH IF NOT
1250  0E69  E0 18           CPX #24         ;SEC IF "1"; CLC IF "0"
1260  0E6B  68              PLA
1270  0E6C  AA              TAX             ;RESTORE X
1280  0E6D  68              PLA             ;RESTORE A
1290  0E6E  60              RTS             ;RETURN
1300  0E6F                ;
1310  0E6F                ;
1320  0E6F                ;
1330  0E6F                ;     SEND BYTE SUBROUTINE
1340  0E6F                ;     THIS SUB SENDS THE BYTE CONTAINED
1350  0E6F                ;     IN THE ACC TO THE TAPE ALONG WITH
1360  0E6F                ;     A START BIT AND ONE STOP BIT.  X AND
1370  0E6F                ;     Y ARE CLEARED, A IS PRESERVED,
1380  0E6F                ;     CARRY IS SET.  RELAY 1 IS TURNED ON,
1390  0E6F                ;     RELAY 2 IS TURNED OFF.
1400  0E6F                ;
1410  0E6F  18           SNDBYT CLC
1420  0E70  20 25 0E            JSR SNDBIT      ;SEND START BIT
1430  0E73  A2 09              LDX #9          ;SET BIT COUNTER TO 9
1440  0E75  38              SEC             ;SET STOP BIT
1450  0E76  2A           NEXT1  ROL A           ;MOVE BIT TO CARRY
1460  0E77  20 25 0E            JSR SNDBIT      ;SEND IT
1470  0E7A  CA              DEX             ;LAST BIT?
1480  0E7B  D0 F9              BNE NEXT1       ;BRANCH IF NOT
1490  0E7D  60              RTS             ;RETURN
1500  0E7E                ;
1510  0E7E                ;
1520  0E7E                ;
1530  0E7E                ;     GET BYTE SUBROUTINE
1540  0E7E                ;     THIS SUB WILL PICK UP A BYTE FROM
1550  0E7E                ;     TAPE AND RETURN IT IN THE ACC.
1560  0E7E                ;     X AND Y ARE CLEARED, CARRY CONTAINS
1570  0E7E                ;     THE STOP BIT.  RELAYS ARE NOT AFFECTED.
1580  0E7E                ;
1590  0E7E  20 49 0E     GETBYT JSR DETBIT      ;LOOK FOR START BIT
1600  0E81  B0 FB              BCS GETBYT      ;BRANCH UNTIL FOUND
1610  0E83  A2 09              LDX #9          ;SET BIT COUNTER TO 9
1620  0E85  2A           NEXT2  ROL A           ;MOVE BIT TO ACC
1630  0E86  20 49 0E            JSR DETBIT      ;GET NEXT BIT
1640  0E89  CA              DEX             ;LAST BIT?
1650  0E8A  D0 F9              BNE NEXT2       ;BRANCH IF NOT
1660  0E8C  60              RTS             ;RETURN
```

```
1670  0E8D                ;
1680  0E8D                ;
1690  0E8D                ;
1700  0E8D                ;       CHECK ADDRESS SUBROUTINE
1710  0E8D                ;       THIS SUB COMPARES THE POINTER TO
1720  0E8D                ;       THE END ADDRESS AND SETS THE CARRY
1730  0E8D                ;       IF THEY ARE THE SAME.  IF THEY ARE
1740  0E8D                ;       NOT, THE CARRY IS CLEARED AND POINTER
1750  0E8D                ;       IS INCREMENTED.  IT ALSO ADDS THE
1760  0E8D                ;       CONTENTS OF THE ACC TO THE CHECKSUM
1770  0E8D                ;       AND DISPLAYS THE LOW BYTE OF POINTER.
1780  0E8D                ;       A IS MUTILATED, X AND Y ARE PRESERVED.
1790  0E8D                ;       RELAYS ARE NOT AFFECTED.
1800  0E8D                ;
1810  0E8D  D8            CHKADD CLD
1820  0E8E  18                    CLC
1830  0E8F  65 EE                 ADC CHKSUM        ;UPDATE CHECKSUM
1840  0E91  85 EE                 STA CHKSUM        ;SAVE IT
1850  0E93  A5 F6                 LDA PNTER         ;GET POINTER LOW
1860  0E95  8D 20 08              STA DISPLY        ;DISPLAY IT
1870  0E98  C5 F2                 CMP ENDADR        ;CMP WITH END ADDR LOW
1880  0E9A  D0 06                 BNE INCPTR        ;BRANCH IF NOT EQUAL
1890  0E9C  A5 F7                 LDA PNTER+1       ;GET POINTER HIGH
1900  0E9E  C5 F3                 CMP ENDADR+1      ;CMP WITH END ADDR HIGH
1910  0EA0  F0 07                 BEQ RET           ;BRANCH AND SEC IF SAME
1920  0EA2  E6 F6           INCPTR INC PNTER         ;INC LOW BYTE
1930  0EA4  D0 02                 BNE SKIP2         ;BRANCH IF NO CARRY
1940  0EA6  E6 F7                 INC PNTER+1       ;INC HIGH BYTE
1950  0EA8  18              SKIP2  CLC
1960  0EA9  60              RET    RTS               ;RETURN
1970  0EAA                ;
1980  0EAA                ;
1990  0EAA                ;
2000  0EAA                ;
2010  0EAA  10 21           CASS   BPL LOAD          ;BR IF COMMAND= "LOAD"
2020  0EAC                ;
2030  0EAC                ;
2040  0EAC                ;
2050  0EAC                ;       DUMP BLOCK SUBROUTINE
2060  0EAC                ;       THIS SUB TRANSFERS A BLOCK OF MEMORY
2070  0EAC                ;       TO TAPE.  BEGINNING OF THE BLOCK IS
2080  0EAC                ;       SPECIFIED WITH "PNTER", END OF THE
2090  0EAC                ;       BLOCK WITH "ENDADR", AND BLOCK
2100  0EAC                ;       IDENTIFICATION WITH "IDENT".  A
2110  0EAC                ;       CHECKSUM IS SENT AT THE END OF
2120  0EAC                ;       THE BLOCK.  A, X, AND Y ARE CLEARED,
2130  0EAC                ;       CARRY IS SET.  RELAY 1 IS TURNED ON,
2140  0EAC                ;       RELAY 2 IS TURNED OFF.
2150  0EAC                ;
2160  0EAC  A2 FF           DUMP   LDX #255          ;SET UP BIT COUNTER
2170  0EAE  38              NEXT3  SEC               ;"1" BIT
2180  0EAF  20 25 0E              JSR SNDBIT        ;SEND LEADER
2190  0EB2  CA                    DEX               ;FINISHED?
2200  0EB3  D0 F9                 BNE NEXT3         ;BRANCH IF NOT
2210  0EB5                ;
2220  0EB5  86 EE                 STX CHKSUM        ;CLEAR CHECKSUM
2230  0EB7  A5 F1                 LDA IDENT         ;GET IDENTIFIER
```

```
2240  0EB9  20 6F 0E          JSR SNDBYT      ;SEND IT
2250  0EBC                  ;
2260  0EBC  A1 F6       NEXT4  LDA (PNTER,X)   ;GET BYTE
2270  0EBE  20 6F 0E          JSR SNDBYT      ;SEND IT
2280  0EC1  20 8D 0E          JSR CHKADD      ;LAST BYTE?
2290  0EC4  90 F6             BCC NEXT4       ;BRANCH IF NOT
2300  0EC6                  ;
2310  0EC6  A5 EE             LDA CHKSUM      ;GET CHECKSUM
2320  0EC8  20 6F 0E          JSR SNDBYT      ;SEND IT
2330  0ECB  8A               TXA             ;CLEAR ACC
2340  0ECC  60               RTS             ;RETURN
2350  0ECD                  ;
2360  0ECD                  ;
2370  0ECD                  ;
2380  0ECD                  ;     LOAD BLOCK SUBROUTINE
2390  0ECD                  ;     THIS SUB WILL SEARCH FOR A BLOCK ON
2400  0ECD                  ;     TAPE WITH AN IDENTIFIER THAT MATCHES
2410  0ECD                  ;     "IDENT".  WHEN FOUND, IT WILL TRANSFER
2420  0ECD                  ;     THE BLOCK FROM TAPE TO MEMORY.
2430  0ECD                  ;     BEGINNING OF THE DESTINATION IS
2440  0ECD                  ;     SPECIFIED WITH "PNTER" AND END WITH
2450  0ECD                  ;     "ENDADR".  FOR THIS REASON THE SIZE
2460  0ECD                  ;     OF THE BLOCK MUST BE KNOWN.  ACC IS
2470  0ECD                  ;     RETURNED WITH "AA" FOR SUCCESSFUL LOAD
2480  0ECD                  ;     AND "EE" FOR ERROR.  X IS CLEARED,
2490  0ECD                  ;     Y IS SMASHED.  RELAYS ARE NOT AFFECTED.
2500  0ECD                  ;
2510  0ECD  A2 14       LOAD   LDX #20         ;SEARCH FOR LEADER
2520  0ECF  20 49 0E    LOOP   JSR DETBIT      ;GET A BIT
2530  0ED2  90 F9             BCC LOAD        ;START OVER IF "0" BIT
2540  0ED4  CA               DEX             ;20 BITS YET?
2550  0ED5  D0 F8             BNE LOOP        ;BRANCH IF NOT
2560  0ED7  20 7E 0E          JSR GETBYT      ;GET IDENTIFIER FROM TAPE
2570  0EDA  8D 20 08          STA DISPLY      ;DISPLAY IT
2580  0EDD  A4 F1             LDY IDENT       ;GET DESIRED ID
2590  0EDF  F0 04             BEQ SKIP3       ;SKIP COMPARISON IF ID=00
2600  0EE1  C5 F1             CMP IDENT       ;CORRECT ID?
2610  0EE3  D0 E8             BNE LOAD        ;START OVER IF NOT
2620  0EE5  86 EE       SKIP3  STX CHKSUM      ;CLEAR CHECKSUM
2630  0EE7                  ;
2640  0EE7  20 7E 0E    NEXT5  JSR GETBYT      ;GET A BYTE
2650  0EEA  90 10             BCC ERROR       ;BRANCH IF STOP BIT=0
2660  0EEC  81 F6             STA (PNTER,X)   ;STORE BYTE BY POINTER
2670  0EEE  20 8D 0E          JSR CHKADD      ;LAST BYTE?
2680  0EF1  90 F4             BCC NEXT5       ;IF NOT, GET NEXT BYTE
2690  0EF3                  ;
2700  0EF3  20 7E 0E          JSR GETBYT      ;GET CHECKSUM
2710  0EF6  A0 AA             LDY #$AA        ;A-OK MESSAGE
2720  0EF8  C5 EE             CMP CHKSUM      ;IS CHECKSUM OK?
2730  0EFA  F0 02             BEQ SKIP4       ;SKIP ERROR MESSAGE IF SO
2740  0EFC  A0 EE       ERROR  LDY #$EE        ;ERROR MESSAGE
2750  0EFE  98           SKIP4  TYA             ;XFER MESSAGE TO ACC
2760  0EFF  60               RTS             ;RETURN
2770  0F00                  ;
2780  0F00                  ;
2790  0F00                  END
2800  0F00                      .END
```

```
ERRORS = 0000


SYMBOL TABLE

 MANUAL  0E19     CASS    0EAA     CYCLE   0E2E     INCPTR  0EA2
 RET     0EA9     SKIP2   0EA8     LOAD    0ECD     SKIP3   0EE5
 ERROR   0EFC     SKIP4   0EFE     SNDBIT  0E25     PORT    0900
 DISPLY  0820     PIEBUG  0F52     BEEP    0F22     STATUS  00EF
 CHKSUM  00EE     PNTER   00F6     COMAND  00F0     IDENT   00F1
 ENDADR  00F2     BEGADR  00F4     START   0E00     DELAY   0E1C
 DLY     0E21     GAP     0E3F     DETBIT  0E49     TONE    0E4C
 COUNT   0E53     CHECK   0E58     SNDBYT  0E6F     NEXT1   0E76
 GETBYT  0E7E     NEXT2   0E85     CHKADD  0E8D     DUMP    0EAC
 NEXT3   0EAE     NEXT4   0EBC     LOOP    0ECF     NEXT5   0EE7
 END     0F00
```

TAPE MOTION CONTROLS

One of the more unique features of the CS-87 POT-SHOT cassette interface system is the built-in hardware and software provision for relays to control tape motion. The relays are to be supplied by the user and it should be understood that when used with simple cassette recorders only start and stop control is possible.

THE RELAYS

Almost any small relays having either 6 volt or 12 volt coils may be used with the tape motion control, with the following two restrictions:

1) The coils should pull no more than 20 ma. of current when activated.
2) The relay must have a set of NORMALLY CLOSED contacts.

As shown in the accompanying illustration, relays with 6 volt coils are to be connected between GND (pin 5) and the appropriate relay control pin of the cassette connector header (pin 7 for relay 1 and pin 8 for relay 2).



Relays with 12 volt coils are connected between the appropriate relay control pin of the cassette connector header and its -9 volt supply point (pin 2).

NOTE that the Normally Closed contacts of the relay are used to control the recorder (so that the recorder will be operational when power to the computer is off).

THE SOFTWARE

The normal operating software for the CS-87 POT-SHOT cassette system controls relay #1 only (except that relay #2 is initialized as being off both by PIEBUG and POT-SHOTs SNDBIT subroutine).

Relay #1 is automatically turned on at the beginning of LOADs and DUMPs and turned off when the tape operation is complete.

To facilitate fast forward and rewind operations of the recorder, a bypass function over the normal operating sequence of the relays has been provided. This by-pass is activated by entering ØØ from the 8700 keyboard followed by touching the TAPE control pad. This operation turns relay #1 on and leaves it on until any key of the 8700 keyboard is touched, at which time the relay is de-activated.

POT-SHOT TAPE FORMAT

POT-SHOT uses a technique that was first described in the September 1975 issue of Popular Electronics as the "HIT" System (Hobbyist Interchange Tape System). The technique is one that uses tone bursts to represent bits on the tape. However, the format has been changed to gain greater reliability. A "0" bit is represented as 8 cycles of 2000 Hz. and a "1" bit is 16 cycles of 2000 Hz. The dead space between bits is a constant 4 milliseconds (same amount of time as a "0" bit takes). We found this technique and format to be the most tolerant of speed variations and distortion produced by the cassette recorder.

The format of a file on tape is as follows: First there is a 3 second leader tone that consists of 255 "1" bits. This leader serves three purposes; it gives the tape recorder time to come up to the speed, the automatic level control time to stabilize (if the recorder has one), and identifies the start of a file. Next the file identifier byte is sent. Then the actual file is sent starting with the byte specified by "BEGINNING ADDRESS" and continuing to the byte specified by "END ADDRESS". After the file, a one byte checksum is put on tape to ensure data integrity when the file is loaded back into the computer.

CASSETTE INPUT PORT

ADDRESS-F900



CASSETTE OUTPUT PORT

ADDRESS-F900



CASSETTE CONNECTOR

J9



Schematics

CASSETTE INPUT PORT



CASSETTE OUTPUT PORT